Estimating noise in op amp stages

Estimating the noise performance of an op amp stage is easy with a little circuit analysis and a short BASIC program to take care of the maths. The program requires only two resistance values and a figure for bandwidth to compute the noise levels for six popular op amps.

by PHIL ALLISON

There are several sources of noise in an op amp stage which together account for the total background hiss level. These are the op amp itself (particularly the active devices employed in the input stage), the resistors used for gain setting, and the noise generated by the resistance of the signal source.

It must be appreciated that any resistor has a self noise level caused by thermal agitation of its free electrons. This noise, commonly known as white noise, is random and spreads across the whole frequency spectrum. Its magnitude is given by a simple formula:

where

En = RMS noise voltage

L = Boltzmann's constant 1.38 x 10^{-23}

T = temperature in degrees K (degrees C + 273)

B = bandwidth of measurement

R = resistor value in ohms

For example: a 10kΩ resistor at room temperature and measured with a 20kHz bandwidth will generate a noise voltage of 1.8μV. (Try some other values on your calculator to get a feel for the quantities involved).

The program presented here can be used to select the best op amp for a given application or to examine the effect on noise performance of design changes to a circuit.

Before the program can be used, two resistance values must be derived from the circuit of the op amp stage in question. These I have called *source resistance* and *input resistance*. The first is just the value in ohms of the internal resistance of the device generating the input signal.

For example, for a 200-ohm microphone use a value of 200 for the source resistance, and for a high impedance microphone (internal step-up transformer type) use a value of 50,000. If noise testing is to be done with the input shorted then use a value of 1 (one ohm) as the program will not accept a value of 0.

Input resistance

The input resistance has to be determined from the circuit of the gain stage in question and here a little analysis is needed. Note that the input resistance is not the same as the input impedance for the circuits of Fig.1 and Fig.2.

There are two common types of op amp gain stages: (1) the inverting stage as shown in Fig.1; and (2) the non-inverting stage as shown in Fig.2. The input impedance of the inverting type is equal to R1, while the input impedance of the non-inverting type is equal to Rin and may be almost any value. The signal gains of these two stages are given by the formulas beneath each diagram.

Don't worry if your circuit has capacitors in series with the input or feedback ground (Fig.2) as normally these can be neglected.

In Fig.1, the input resistance is equal to R1 in parallel with R2. If R2 is more than ten times R1, then just use the value of R1.

For Fig.2, the input resistance is the same as for Fig.1 (ie, R1 in parallel R2), but if Rin is less than ten times R1 then calculate Rin in parallel with R1 and R2 as well. If there is a resistor in series with the input, add this to the input resistance.

The figure for bandwidth can be any value up to the circuit bandwidth. For audio purposes, a figure of about 16kHz is commonly adopted for specifications.

The program will, in a couple of seconds, compute the *equivalent input noise* (EIN) and noise figure for six op amps. Other op amps can easily be added to the list.

The EIN is a standard way of specifying input stage noise as it is independant of the overall gain. If you multiply the EIN figure by the gain of the stage, then you will have the noise voltage expected at the output.

The noise figure is also calculated so that the standard of performance of a circuit can be seen at a glance. It compares the stage in question with an imaginary noiseless stage and quotes the difference in decibels. A figure of 1dB would be very good and hardly worth trying to improve upon. This figure is



Fig.1: inverting op amp stage. Gain = R2/R1.



Fig.2: non-inverting op amp stage. Gain = (R1 + R2)/R1.

```
10 CLS:PRINT
20 PRINT"   PROGRAM TO CALCULATE NOISE"
25 PRINT"          IN OP AMPS"
30 PRINT"   **************************"
40 PRINT
50 INPUT" SOURCE RESISTANCE   ";RS:PRINT:IFRS=0THEN50
60 INPUT" INPUT RESISTANCE    ";RI:PRINT:IFRI=0THEN60
70 INPUT" NOISE BANDWIDTH KHZ ";BW:PRINT:IFBW=0THEN70
71 PRINT
100 DATA 3.5E-9,4E-13,1E-8,5E-13,1.8E-8,1E-14
110 DATA 1.5E-8,1.7E-13,2.2E-8,6E-13,4.7E-8,1E-14
115 RESTORE
120 FORI=1TO6:READ EN,IN
140 KT=4.1E-21
150 ET=((EN^2+IN^2*(RS^2+RI^2)+4*KT*(RS+RI))*BW*1E3)^0.5
160 IFI=1THENPRINT"   NE5534 ";:GOTO300
170 IFI=2THENPRINT"   RC4558 ";:GOTO300
180 IFI=3THENPRINT"   TL071  ";:GOTO300
190 IFI=4THENPRINT"   LM301A ";:GOTO300
200 IFI=5THENPRINT"   UA741C ";:GOTO300
202 IFI=6THENPRINT"   TL081  ";:GOTO300
300 PRINTUSING"###.##";ET*1E6;:PRINT" UV ";
310 NS=(4*KT*BW*1E3*RS)^0.5
320 NF=20*LOG((ET/NS))/LOG(10)
330 PRINTUSING" ##.#";NF;:PRINT"  DB"
340 NEXTI
350 PRINT" ============================"
360 INPUT"RTN";A:IFA=0SOUND21,1:GOTO10
```

Left: this program was written for the VZ300 computer but should work with little alteration on almost any computer running BASIC. The program runs each time return is pressed, so that you can enter new values.

Below: these sample screen printouts show the results for six common op amps for various circuit conditions. The program calculates both the equivalent input noise (in microvolts) and the noise performance (in dB).

```
S A M P L E   S C R E E N S

SOURCE RESISTANCE   ? 200

INPUT RESISTANCE    ? 47

NOISE BANDWIDTH KHZ ? 16

  NE5534    0.51 UV    7.0  DB
  RC4558    1.29 UV   15.0  DB
  TL071     2.29 UV   20.0  DB
  LM301A    1.91 UV   18.4  DB
  UA741C    2.79 UV   21.7  DB
  TL081     5.95 UV   28.3  DB
============================

SOURCE RESISTANCE   ? 7000

INPUT RESISTANCE    ? 1000

NOISE BANDWIDTH KHZ ? 16

  NE5534    1.56 UV    1.2  DB
  RC4558    1.97 UV    3.3  DB
  TL071     2.70 UV    6.0  DB
  LM301A    2.39 UV    4.9  DB
  UA741C    3.18 UV    7.4  DB
  TL081     6.12 UV   13.1  DB
============================

SOURCE RESISTANCE   ? 1E5

INPUT RESISTANCE    ? 1E4

NOISE BANDWIDTH KHZ ? 2.5

  NE5534    2.93 UV    3.2  DB
  RC4558    3.33 UV    4.3  DB
  TL071     2.31 UV    1.1  DB
  LM301A    2.41 UV    1.5  DB
  UA741C    3.85 UV    5.6  DB
  TL081     3.17 UV    3.9  DB
============================
```

independent of gain, bandwidth and signal level.

Low noise tips

To optimise a design, the value of input resistance must be kept as low as possible. For an inverting stage, this is limited by the minimum acceptable input impedance. There is no such problem with the non-inverting stage, making it the preferred type for low noise stages. Most op amps will drive loads down to 1000 ohms or so, hence R1 plus R2 can equal this. The NE5534 can drive loads down to 600 ohms.

Don't worry about using expensive "low noise" resistors as these make no difference in an op amp stage where there is little or no DC across the resistors. Noise caused by a large voltage across a resistor is called excess noise and varies widely with resistor type.

Using the program

The formula for noise in the program appears in line 150. This sums all the noise sources involved using the published data for each op amp in turn and the result is quoted in microvolts. This data appears in lines 100 and 110 as EIN voltage and EIN current figures in volts and amps per Hz respectively. Line 320 computes the noise figure by dividing the result of line 150 by the noise of the source resistance and converting this to decibels.

When return is pressed the program runs again so that you can enter new values.

Due to device variations and the use of averaged values in the EIN data, the computed figures are not precise but are close enough to measured results to allow valid comparisons between circuits and op amps.

The program was written for a VZ300 computer but should work with little alteration on almost any computer running BASIC.

References

R.A. Fairs, Resistor Survey. *Wireless World*, October 1975.

Walter G. Jung, *IC Op Amp Cookbook*.